
Sistema de modelado algebraico Características generales

Sistemas de modelado algebraico

En general disponen de un lenguaje de modelado algebraico que permite
trabajar en alto nivel de abstracción, utilizando una notación similar a la
matemática.

Algunos tienen un entorno de desarrollo integrado (IDE) y/o un interprete de
línea de comandos incorporados.

Todos proveen mecanismos para vincularlos con solvers y así poder resolver
los modelos. En algunos casos el solver viene incluido en el sistema de
modelado.

Ejemplos:

GAMS – General Algebraic Modeling System

Xpress – Modelado matemático y optimización

AMPL – A Modeling Language for Mathematical Programming

GLPK – GNU Linear Programming Kit

Sistema de modelado algebraico GNU Linear Programming Kit

GLPK

El GNU Linear Programming Kit (GLPK) es un sistema de modelado
algebraico con soporte para problemas de programación entera mixta. Es
software libre con licencia GPL.

Como lenguaje de modelado algebraico utiliza GNU MathProg que es una
versión libre pero reducida del lenguaje de modelado algebraico AMPL,
considerado un estándar de echo para problemas de optimización matemática

Incluye un solver para problemas de programación lineal o entera mixta, que
puede ser invocado por línea de comandos mediante el comando glpsol
para resolver modelos escritos en MathProg.

Sistema de modelado algebraico Instalación

Instalación en Ubuntu y macOS

GLPK está disponible en una gama amplia de sistemas operativos.

Sitio oficial (código fuente, listas de correo):
http://www.gnu.org/software/glpk/glpk.html

Ubuntu:
sudo apt install glpk-utils

macOS:

Utilizando homebrew: brew install glpk

Utilizando macports: sudo port install glpk

Sistema de modelado algebraico Instalación

Instalación en Windows

GLPK for Windows:
http://winglpk.sourceforge.net/

El proyecto GLPK for Windows provee ejecutables pre-compilados para
Windows.

El zip de la distribución contiene los ejecutables para 32 y 64 bits en las
carpetas w32 y w64 respectivamente. Se recomienda agregar la carpeta
correspondiente al PATH del sistema para poder invocar al solver desde
cualquier carpeta.

GUSEK (GLPK Under Scite Extended Kit):
http://gusek.sourceforge.net/gusek.html

GUSEK provee un IDE básico para GLPK. Incluye un editor de texto con
resaltado de syntaxis para los archivos de modelo y datos, consola de errores
integrada y la posibilidad ejecutar el solver desde la interfaz gráfica.

Sistema de modelado algebraico Instalación

Verificación

Para verificar instalación, ejecutar desde línea de comandos:

glpsol --version

Salida esperada:

GLPSOL: GLPK LP/MIP Solver, v4.55

Copyright (C) 2000, 2001, 2002, 2003, 2004, 2005, 2006, 2007, 2008,
2009, 2010, 2011, 2013, 2014 Andrew Makhorin, Department for Applied
Informatics, Moscow Aviation Institute, Moscow, Russia. All rights
reserved. E-mail: <mao@gnu.org>.

This program has ABSOLUTELY NO WARRANTY.

This program is free software; you may re-distribute it under the terms
of the GNU General Public License version 3 or later.

Sistema de modelado algebraico Documentación GNU MathProg

Documentación GNU MathProg

Las distintas distribuciones tienen incluida la documentación de GLPK.

En particular, el manual de referencia de MathProg es el archivo gmpl.pdf.

También se puede encontrar online, por ejemplo en:
http://gusek.sourceforge.net/gmpl.pdf

Sistema de modelado algebraico Uso básico

Formulación básica (vectorial)

max cτx
s.a Ax ≤ b

x ≥ 0, entero.

Sistema de modelado algebraico Uso básico

Formulación básica (extendida)

max cτx
s.a Ax ≤ b

x ≥ 0, entero.

max
∑

j∈J cjxj

s.a
∑

j∈J Aijxj ≤ bi,∀i ∈ I
xj ≥ 0, entero, ∀j ∈ J.

Sistema de modelado algebraico Uso básico

Formulación básica (MathProg)

max
∑

j∈J cjxj

s.a
∑

j∈J Aijxj ≤ bi,∀i ∈ I
xj ≥ 0, entero, ∀j ∈ J.

maximize obj:
sum{j in J} c[j] * x[j];

s.t. ctr{i in I}:
sum{j in J} A[i,j] * x[j] <= b[i];

Sistema de modelado algebraico Uso básico

Formulación básica (MathProg modelo completo)

Conjuntos
I
J

Parámetros
cj

bi

Aij

Variable de decisión
xj

max
∑

j∈J cjxj

s.a
∑

j∈J Aijxj ≤ bi,∀i ∈ I
xj ≥ 0, entero, ∀j ∈ J.

set I;
set J;

param c{J};
param b{I};
param A{I,J};

var x{J} >= 0, integer;

maximize obj:
sum{j in J} c[j] * x[j];

s.t. ctr{i in I}:
sum{j in J} A[i,j] * x[j]
<= b[i];

end;

Sistema de modelado algebraico Uso básico

Formulación básica (modelo y datos)

basico.mod basico.dat

set I;
set J;

param c{J};
param b{I};
param A{I,J};

var x{J} >= 0, integer;

maximize obj:
sum{j in J} c[j] * x[j];

s.t. ctr{i in I}:
sum{j in J} A[i,j] * x[j] <= b[i];

end;

set I := 1, 2, 3;
set J := 1, 2, 3, 4;

param c := 1 11,
2 22,
3 33,
4 44;

param b :=
1 100, 2 200, 3 300;

param A :
1 2 3 4 :=

1 5 6 7 8
2 9 10 11 12
3 13 14 15 15 ;

end;

Sistema de modelado algebraico Uso básico

Resolución

Ejecutar desde línea de comandos:

glpsol --model basico.mod --data basico.dat --output basico.out

--model indica el archivo con el modelo

--data indica el archivo con los datos

--output indica el archivo para guardar la solución

Sistema de modelado algebraico Uso básico

Salida (basico.out)

Resumen con tamaño del problema y status final:

Problem: basico
Rows: 4
Columns: 4 (4 integer, 0 binary)
Non-zeros: 16
Status: INTEGER OPTIMAL
Objective: obj = 528 (MAXimum)

Sistema de modelado algebraico Uso básico

Salida (basico.out)

Actividad en las filas:

No. Row name Activity Lower bound Upper bound
------ ------------ ------------- ------------- -------------

1 obj 528
2 ctr[1] 100 100
3 ctr[2] 152 200
4 ctr[3] 195 300

Actividad en las columnas:

No. Column name Activity Lower bound Upper bound
------ ------------ ------------- ------------- -------------

1 x[1] * 0 0
2 x[2] * 0 0
3 x[3] * 4 0
4 x[4] * 9 0

Sistema de modelado algebraico Uso básico

Modelo y datos en un mismo archivo

todo.mod

set I;
set J;

maximize obj: ...

data;

set I := 1, 2, 3;
set J := 1, 2, 3, 4;

...

end;

En este caso no hace falta especificar --data al resolver:
glpsol --model todo.mod --output todo.out

Sistema de modelado algebraico Ejemplos de modelado

Ejemplo: Mochilero

mochilero.mod mochilero.dat

param n;
set I := 1 .. n; # objetos

param c{I}; # valor
param a{I}; # tamaño
param b; # capacidad

var x{I} binary; # llevar si/no

maximize valor:
sum{i in I} c[i] * x[i];

s.t. capacidad:
sum{i in I} a[i] * x[i] <= b;

end;

param n := 8;

param c :=
1 1, 2 10, 3 15,
4 40, 5 60, 6 90,
7 100, 8 15;

param a :=
1 10, 2 60, 3 30,
4 40, 5 30, 6 20,
7 20, 8 2;

param b := 101;

end;

Sistema de modelado algebraico Ejemplos de modelado

Ejemplo: Asignación

asignacion.mod asignacion.dat

param n;
set I := 1 .. n; # personas
set J := 1 .. n; # tareas

param c{I,J}; # costo

var x{I,J} binary; # asignación i a j

minimize costo:
sum{i in I, j in J} c[i,j] * x[i,j];

s.t. tareas_personas{i in I}:
sum{j in J} x[i,j] = 1;

s.t. personas_tareas{j in J}:
sum{i in I} x[i,j] = 1;

end;

param n := 4;

param c :
1 2 3 4 :=

1 54 54 51 53
2 51 57 52 52
3 50 53 54 56
4 56 54 55 53 ;

end;

Sistema de modelado algebraico Ejemplos de modelado

Ejemplo: Conjuntos indizados y parámetros de más de dos dimensiones

prod.mod prod.dat

set P; # productos
set A{P}; # áreas de mercado
param h; # número semanas
set T := 1..h; # semanas

# límite de producto por semana
param market{p in P, A[p], T}
>= 0;

# producto vendido
var Sell{
p in P, a in A[p], t in T

} >= 0, <= market[p,a,t];

...

end;

param h := 3;
set P := b c;
set A[b] := east north;
set A[c] := east west export;

param market :=
[b,*,*]: 1 2 3 :=
east 2000 2000 1500
north 4000 4000 2500

[c,*,*]: 1 2 3 :=
east 1000 800 1000
west 2000 1200 2000
export 1000 500 500 ;

...

end;

Sistema de modelado algebraico Ejemplos de modelado

Ejemplo: Producto cartesiano en vez de conjunto indizado

prod2.mod prod2.dat

set P; # productos
set A; # áreas de mercado
set PA within P cross A; # <-
param h; # número semanas
set T := 1..h; # semanas

# límite de producto por semana
param market{PA, T} >= 0;

# producto vendido
var Sell{(p, a) in PA, t in T}
>= 0, <= market[p,a,t];

...

end;

param h := 3;
set P := b c;
set A :=
east west north export;

set PA :=
(b,east) (b,north)
(c,east) (c,west) (c,export);

param market :=
[b,*,*]: 1 2 3 :=
east 2000 2000 1500
north 4000 4000 2500

[c,*,*]: 1 2 3 :=
east 1000 800 1000
west 2000 1200 2000
export 1000 500 500 ;

Sistema de modelado algebraico Salida personalizada

Impresión de resultados

MathProg incluye comandos para imprimir el resultado de la evaluación de
expresiones, tanto en la salida estándar como en archivo.

Los comandos disponibles son:

display

printf

A la hora de imprimir los resultados nos va a resultar útil iterar sobre los
datos. El comando disponible para iteración en MathProg es for.

Sistema de modelado algebraico Salida personalizada

Impresión de resultados

Esto es particularmente útil para ver en forma más amigable el resultado de la
resolución del modelo. Para esto es requerido que previamente se invoque la
resolución del modelo con el comando solve.

modelo.mod
set I;
set J;

maximize obj: ...

solve;

... display, printf, for ...

end;

Sistema de modelado algebraico Salida personalizada

display

Modelo Salida

param n;
set I := 1 .. n; # objetos

param c{I}; # valor
param a{I}; # tamaño
param b; # capacidad

var x{I} binary; # llevar si/no

... modelo mochilero ...

solve;

display{i in I}: i, x[i];
display:
sum{i in I} a[i] * x[i], b;

end;

Display statement at line 18
i = 1
x[1].val = 1
i = 2
x[2].val = 0
i = 3
x[3].val = 0
i = 4
x[4].val = 0
i = 5
x[5].val = 1
i = 6
x[6].val = 1
i = 7
x[7].val = 1
i = 8
x[8].val = 1
Display statement at line 19
82
b = 101

Sistema de modelado algebraico Salida personalizada

printf

Modelo Salida

param n;
set I := 1 .. n; # objetos

param c{I}; # valor
param a{I}; # tamaño
param b; # capacidad

... modelo mochilero ...

solve;

printf{i in I}:
"objeto %s, decisión %d\n", i, x[i];

printf:
"volumen %.2f, \ncapacidad %d\n",
sum{i in I} a[i] * x[i], b;

end;

objeto 1, decisión 1
objeto 2, decisión 0
objeto 3, decisión 0
objeto 4, decisión 0
objeto 5, decisión 1
objeto 6, decisión 1
objeto 7, decisión 1
objeto 8, decisión 1
volumen 82.00,
capacidad 101

Sistema de modelado algebraico Salida personalizada

for

Modelo Salida

param n;
set I := 1 .. n; # objetos

... modelo mochilero ...

solve;

for{i in I : x[i] == 1} {
printf: "El objeto %s es llevado\n",
i;

}

for{i in I : x[i] == 0} {
printf: "El objeto %s", i;
printf: " NO es llevado\n";

}

end;

El objeto 1 es llevado
El objeto 5 es llevado
El objeto 6 es llevado
El objeto 7 es llevado
El objeto 8 es llevado
El objeto 2 NO es llevado
El objeto 3 NO es llevado
El objeto 4 NO es llevado

Sistema de modelado algebraico Salida personalizada

Ejecución

Por omisión los mensajes se van a mostrar en la consola:

glpsol --model prob.mod --data prob.dat

Para guardar la salida personalizada en un archivo, se debe utilizar la opción
--display:

glpsol --model prob.mod --data prob.dat --display prob.dsp

